


Overview, Design concepts, and Details + Decision making (ODD+D) for the instantiation of a model for a work environment and workers with Two-Factor Theory.

As the first ABM to apply and test Herzberg's et al. (1959) full TFT, the HRM-PET model presented in this paper provides a bridge for agent-based modeling methods into the HRM literature and, thus, provides a unique contribution to both the HRM and simulation fields. HRM-PET captures each of the heterogeneous motivation factors and heterogeneous hygiene factors as outlined by Herzberg et al. (1959), thus allowing for local agent-environment, and agent-agent interaction that leads to emergent organizational-level traits.

Here I introduce the HRM-PET model, built in the highly user-accessible open source software NetLogo (Wilensky, 1999), Agent-based modeling allows us to study a system (i.e., the workplace, in this application), emerging from individual agents (i.e., workers) interacting with an environment (i.e., work units). Important aspects relating to this system, its agents, and their environment will be introduced below. The following description follows Grimm, et al.'s ODD protocol (Grimm et al. 2006, 2010). The rationale for utilizing the ODD and for sharing the model is that it allows broader dissemination of the model and its methodology. Providing this information and model will aid future scholars in replicating the results presented here, and in extending the model if they so desire.

Figure 1. HRM-PET Graphical User Interface



The HRM-PET is designed to be calibrated to a common state of an organization and allow examination of the outputs. However, that organizations face crises or some other fundamental structural change is inevitable (Lin, et al, 2006). Flexible base models, such as the one presented here, allow for the practitioner or researcher to examine the same model using

a variety of calibrations to explore and possibly mitigate crisis impact. Providing this information and model will aid future scholars in replicating the results presented here, and in extending the model if they so desire.

HRM-PET's agents are workers. They have heterogeneous proclivities toward the different motivation factors and heterogeneous tolerance of the different hygiene factors. Based on these proclivities and tolerance levels for each, the agents carry out a series of assessments about their (dis)satisfaction with their work unit. Work units possess corresponding hygiene styles and potentials to enable motivation. Furthermore, the agents have a maximum of two work unit moves in a time-step before they assess their satisfaction and dissatisfaction. The time-steps in the simulation are notional, but abstractly correspond to the time period between received job offer and a job acceptance/refusal from candidate.

Table 1. Motivation and Hygiene Factors as HRM-PET Attributes

Factor	Definition	Related Cell Attributes	Related agent attributes
<i>Motivation</i>			
Achievement	Successful completion of a task, seeing results of own work, solutions to problems	achievement-potential	achievement-style
Recognition	An act of positive notice or praise	recognition-potential	recognition-style
Work Itself	Positive sentiment from the actual tasks required of the job	work-itself-potential	work-itself-style
Responsibility	Responsibility over own work or that of others	responsibility-potential	responsibility-style
Advancement	Upward change in status or position	advancement-potential	advancement-style

<i>Hygiene</i>			
Policy and Administration	The organization's rules for managing the work tasks and the personnel.	policy-style	policy-tolerance
Supervision-Technical	Level of competence and fairness of the employee's supervisor.	supervision-style	supervision-tolerance
Salary	Quality of compensation.	salary-style	salary-tolerance
Interpersonal Relations-Supervisor	Quality of interactions with employee's peers, subordinates, and supervisors.	relationship-style	relationship-tolerance
Work Conditions	Quality of physical conditions of the work or the facilities, and quantity of work.	conditions-style	conditions-tolerance

The initial factor level distribution within the population is set by the model's experimenter, who may choose a uniform ($0 > 10$) or normal ($\mu = 5, \sigma = 2.5$). These allow the user to account for industries that are relatively uniform in their distribution of quality work environments, those that have a meaningful average quality, and those that have only very few high-quality work environments and a large number of low quality work environments. Each agent begins with 0 level satisfaction and dissatisfaction. Once the agents' needs are satisfied (i.e., their hygiene or motivation needs are at least minimally met), or their two possible moves are exhausted, they assess their satisfaction and dissatisfaction.

As in the TFT, the motivation factors contribute to satisfaction while the hygiene factors contribute to dissatisfaction. If a given agent's motivation factor need is less than the potential for that motivation factor within their work unit, the agent's satisfaction increases. Similarly, if

an agent's hygiene factor tolerance is less than the corresponding hygiene factor within the work unit, the agent's dissatisfaction increases. The contribution to increases are pegged to the empirical results Herzberg et al. (1959) originally reported and shown in Table 1. So, too, the HRM-PET preserves the TFT's ratio between factors (e.g., in the HRM-PET, the achievement motivation factor has twice the contribution to job satisfaction as advancement, consistent with Herzberg et al.'s [1959] empirical results).

The description below provides the ODD for model elements that are introduced for the calibration process described in Chapter 4. However, the verification and internal validation presented in Section 3.3 addresses the baseline model. The verification and external validation of the model based on real-world organization data is presented in Chapter 4, Section 4.5.

1. Variables

Given the purpose of the model, the key elements are the work environment (work units) and the workers. The work environment defines the abstract space in which the workers make decisions about satisfaction and the work unit responds to those decisions based on certain variables. A list of variables is provided in Table 2.

Table 3. Variables within HRM-PET

Variable	Worker, Work Unit, Global	Dependent or Independent	Theoretical Grounds/Empirical Basis/Assumptions
achievement-potential	Work Unit	I	Herzberg, et al. 1959
achievement-style	Worker	I	Herzberg, et al. 1959; Shapira, 1989; Gubler, et al., 2016; Lundberg, et al., 2009
recognition-potential	Work Unit	I	Herzberg, et al. 1959
recognition-style	Worker	I	Herzberg, et al. 1959 Shapira, 1989; Gubler, et al., 2016; Lundberg, et al., 2009
work-itself-potential	Work Unit	I	Herzberg, et al. 1959
work-itself-style	Worker	I	Herzberg, et al. 1959; Shapira,

			1989; Gubler, et al., 2016; Lundberg, et al., 2009
responsibility-potential	Work Unit	I	Herzberg, et al. 1959
responsibility-style	Worker	I	Herzberg, et al. 1959; Shapira, 1989; Gubler, et al., 2016; Lundberg, et al., 2009
advancement-potential	Work Unit	I	Herzberg, et al. 1959
advancement-style	Worker	I	Herzberg, et al. 1959; Shapira, 1989; Gubler, et al., 2016; Lundberg, et al., 2009
policy-style	Work Unit	I	Herzberg, et al. 1959
policy-tolerance	Worker	I	Herzberg, et al. 1959; Shirom, et al., 2010; Waddimba, et al., 2016
supervision-style	Work Unit	I	Herzberg, et al. 1959
supervision-tolerance	Worker	I	Herzberg, et al. 1959; Shirom, et al., 2010; Waddimba, et al., 2016
salary-style	Work Unit	I	Herzberg, et al. 1959
salary-tolerance	Worker	I	Herzberg, et al. 1959; Shirom, et al., 2010; Waddimba, et al., 2016

relationship-style	Work Unit	I	Herzberg, et al. 1959
relationship-tolerance	Worker	I	Herzberg, et al. 1959; Shirom, et al., 2010; Waddimba, et al., 2016
conditions-style	Work Unit	I	Herzberg, et al., 1959
conditions-tolerance	Worker	I	Herzberg, et al. 1959; Shirom, et al., 2010; Waddimba, et al., 2016
motivation-weight	Worker	I	Herzberg, et al., 1959
hygiene-weight	Worker	I	Simon, 1955; 1969
Satisfaction	Worker	D	Herzberg, et al., 1959
dissatisfaction	Worker	D	Herzberg, et al., 1959
die-count (agent departure)	Global	D	Poulston, 2009
hire-count	Global	D	Assumed new hire for each agent departure based

1.2 Global Variables

The variables that are tracked on a higher-level macro-scale and represent overall simulation metrics are global variables. One of the global variables tracked within the model is turnover. A tally of workers departing the organization entirely and workers hired to replace them are maintained at the global level. These departures are based on dissatisfaction rather than the agent's time in the simulation. Therefore, they correspond to workers departing an organization based on choices other than retirement. The two turnover variables are the overall count of all departures during a given run of the simulation – die-count – and the count of departures in each time-step. Finally, there is also a count of new hires – sprout-count - filling the role of departed workers, this is largely meant for verification purposes to assure the simulation is running properly.

globals [

```
s-greater-d  
d-greater-s  
sprout-count  
die-count  
step-die-count]
```

Additionally, the mean and standard deviation of both satisfaction and dissatisfaction are monitored as are the number of work units with no workers present, the number of work units with any workers present, the number of work units with greater than 10 workers and greater than 20 workers. These global monitors allow the experimenter to examine the distribution of the labor force.

1.3 The Work Environment

The work environment is a torus-shaped space of either 1225 cells, 225 cells, or 1 cell. Torus-shaped avoids edge of world limitations that creates simulation artifacts unrepresented in the world (e.g. avoids border work units having fewer neighboring work units than non-bordering ones). The 1225 cell and 225 cell allow researchers to test large organization and smaller organizations. The single cell version is both allows for testing of single work unit enterprises and for verification of effects caused by the TFT operation in the model. These representations are discussed further in Section 3.2.4.

Each cell represents a work unit. Each work unit owns a value representing each of the five hygiene factors and a value representing the highest potential level for motivation factors that the work unit is able to accommodate. The factor attributes are heterogeneously applied to the work units based on the initial distribution set by the researcher.

Within the model, the work unit's environmental factors are termed a hygiene-style. Their potential to support worker motivation preferences are motivation-potential. The NetLogo code is as follows:

```
patches-own [  
  policy-style  
  supervision-style  
  relationship-style  
  conditions-style  
  salary-style  
  achievement-potential  
  recognition-potential  
  work-itself-potential  
  responsibility-potential  
  advancement-potential]
```

The work units also maintain monitors of the sum of all their hygiene scores and the total number of workers currently in their work unit, respectively:

```
hygiene-total  
work-unit-population
```

1.4 The workers

The workers begin the simulation with heterogenous levels of each of the motivation factors, representing their motivation style, and heterogenous levels of tolerance for each of the hygiene factors encountered in their work unit. Based on their levels for the factors, they weigh their satisfaction and dissatisfaction and make choices about whether to move work units or leave the organization entirely. Similar to the work units, the initial distributions of these factors among the workers is set by the researcher to be normally, uniformly or Poisson distributed. This allows the user to test the application of different theories of motivational tendencies.

Within the model, the worker's motivation preferences are termed a motivation-style. Their hygiene preferences are termed a hygiene-style. As follows:

```
turtles-own [  
  achievement-style  
  recognition-style  
  work-itself-style  
  responsibility-style  
  advancement-style  
  policy-tolerance  
  supervision-tolerance  
  relationship-tolerance  
  conditions-tolerance  
  salary-tolerance]
```

The workers each have the dependent variables satisfaction and dissatisfaction which is based on the interaction between their own factor preferences and that enabled by their work unit environment.

```
satisfaction  
dissatisfaction
```

They also each have a monitor variable denoting whether they've moved work units in that particular time-step.

```
moved?
```


Finally, the worker's are also imbued with a preference for short- or long-term motivation factors or individual motivation factors in order for the model to operate under different motivation schemes described below.

halfinclination
fifthinclination

2 Process Overview and Scheduling

The workers have a maximum of two work unit moves per time-step based first on whether the initial work unit meets the worker's minimum hygiene tolerance. The ability to and limitation of moves replicates a worker's ability to leave an intolerable job situation while not being able to do so repeatedly in a limited amount of time. Each worker has the opportunity to move work units twice in a given time step, first based on their hygiene requirements and second based on their motivation needs.


From a behavioral economics perspective, the hygiene tolerance levels and motivation factor levels are the boundaries that define what is satisfactory and what is not (à la Simon, 1955). As such, hygiene tolerance may be interpreted in an economic sense as the opportunity cost level, at which point work unit hygiene below this level are unacceptable and the work unit remains acceptable at any level above the workers' tolerance regardless of how far above the tolerance level it is. Whereas motivation factor levels may be interpreted in the psychological sense as aspiration levels, the minimum viable solution.

It is within this bounded rationality of the knowledge of the worker's own requirements and the factors within the worker's own work unit that this decision is made. Simon (1955) proposed the choice model where in sequential presentation of alternatives, the first alternative that meets the selection criteria will be chosen, this was later termed "satisficing (Simon, 1996)." In my model, the worker uses a satisficing behavior (chosen by the researcher) and their own preference to prioritize long term versus short term motivational needs (also chosen by the researcher). With regard to the former process, the possible satisficing behaviors available to apply across the entire worker population are: low-, mid-, or high-satisficing. Under low satisficing conditions, only one hygiene level needs to be met, under mid-satisficing, the policy and administration factor and one other factor needs to be met, under high demand conditions, all minimum hygiene factors need to be met for a worker to choose to stay in their work unit.

Once a worker has exhausted its opportunity to move work units in a time step, it then assess its satisfaction and dissatisfaction levels. In the base version of this theoretical model, the satisfaction and dissatisfaction of workers increases based on the original contribution of each factor towards satisfaction and dissatisfaction as observed in Herzberg et al (1959). However, with known organization relationships between factors and (dis)satisfaction, the level of increase may be calibrated to those levels. If dissatisfaction rises to a level of beyond 100, the agent leaves the cell space to replicate leaving the work force entirely.

The above decisions are made by workers in each timestep and represented in Figure 6. Model time steps are abstract but correspond to the length of time for a worker to move through up to two work units in an organization.

Figure 2: Time Step Operations



3 Design Overview

Table 2: Design Considerations (Reference in Text)

Feature	Utilization
Emergence	<p>Macro: Given that the workers are only determining their own satisfaction and dissatisfaction with a single work unit randomly assigned, the pattern of worker convergence on a set of work units is an emergent bottom-up organizational structure.</p> <p>Micro: Within the agents themselves, the only factors directly contributing to dissatisfaction are hygiene factors. However, motivation-weight parameters have an indirect effect on dissatisfaction scores that arises in the agent-cell interaction.</p>

Adaptation	The workers adapt to intolerable hygiene-levels by leaving one work unit for a neighboring one.
Fitness	The fitness assessment of each worker is their primary function within the model. While this assessment is cumulative within a worker's time within a single work-unit, future iterations of the model could explore the addition of memories of previous work units.
Prediction	The agents in the current model version only base their prediction for future status on a "grass is always greener" approach. They assume that if their tolerance levels for hygiene are not met currently, they are better off in a new work unit.
Sensing	Workers can sense the variance in their motivation requirements and hygiene tolerances and that of the potential and style of their work unit.
Interaction	The workers directly interact with their work units in their own satisfaction/dissatisfaction assessments. They indirectly interact with their fellow work unit workers through the derivation of the work unit's adaptation from the average of all the work unit's workers.
Stochasticity	Multiple sub-models rely on stochastic processes, worker moves to a new work unit are stochastic.
Collectives	Collectives emerge as workers converge on particular work units.
Observation	The model collects data at each time step on mean satisfaction and mean dissatisfaction, standard deviation of satisfaction and dissatisfaction, number of work units populated, number of work units that are large (defined as >20 workers), number of workers that have moved due to not having hygiene tolerance levels met, number of workers that have moved due to not having motivation needs met, number of workers who have left the organization, number of workers hired.

4 Details

4.1 Initialization

The model initializes based on experimenter settings. The possible cell-space sizes are 35 x 35, 15 x 15, or 1 x 1. This variation allows testing of small, medium, and large organizations.

```
to setup
  clear-all
  if Org-size = "35x35" [
    resize-world -17 17 -17 17
    set-patch-size 15
  ]
  if Org-size = "15x15" [
    resize-world -7 7 -7 7
    set-patch-size 25
  ]
  if Org-size = "1x1" [
    resize-world 0 0 0 0
    set-patch-size 100
  ]
]
```

Initial work unit monitors are zeroed.

```
ask patches [
  set satisfaction-turtles-here 0
  set dissatisfaction-turtles-here 0
  set work-unit-population 0
]
```

The factor levels are set based on the form of the distribution they are to take across the whole organization. The hygiene-factor-distribution determines how the work unit's hygiene-styles and motivation-potentials are distributed. The hygiene and motivation factors set at initialization, with the exception of relationship-style, remain the same throughout the simulation. The relationship-style of the work unit changes based on the average relationship-tolerance of the workers present in the work unit during a time-step. This captures changes in supervision based on employee changes.

```
if hygiene-factor-distribution = "random" [
  set policy-style random 10
  set supervision-style random 10
  set relationship-style random 10
  set conditions-style random 10
  set salary-style random 10
  set achievement-potential random 10
  set recognition-potential random 10
  set work-itself-potential random 10
  set responsibility-potential random 10
]
```

```

    set advancement-potential random 10
]
if hygiene-factor-distribution = "normal" [
  set policy-style random-normal 5 2.5
  set supervision-style random-normal 5 2.5
  set relationship-style random-normal 5 2.5
  set conditions-style random-normal 5 2.5
  set salary-style random-normal 5 2.5
  set achievement-potential random-normal 5 2.5
  set recognition-potential random-normal 5 2.5
  set work-itself-potential random-normal 5 2.5
  set responsibility-potential random-normal 5 2.5
  set advancement-potential random-normal 5 2.5
  reset-negatives-patches
]

```

The hygiene factor levels are then summed and the work unit patches take on color scaled based on the sum. Darker cyan denotes better hygiene levels.

```

    set hygiene-total (policy-style + supervision-style +
relationship-style + conditions-style + salary-style)
    set pcolor scale-color cyan hygiene-total 50 0

```

To avoid artifacts of the uncommon but existing possibility of negative numbers arising in NetLogo's distribution in the normal option, I use an operation that turns the negative numbers into 0's.

```

to reset-negatives-patches
  if policy-style < 0 [
    set policy-style 0]
  if supervision-style < 0 [
    set supervision-style 0]
  if relationship-style < 0 [
    set relationship-style 0]
  if conditions-style < 0 [
    set conditions-style 0]
  if salary-style < 0 [
    set salary-style 0]
  if achievement-potential < 0 [
    set achievement-potential 0]
  if recognition-potential < 0 [
    set recognition-potential 0]
  if work-itself-potential < 0 [
    set work-itself-potential 0]
  if responsibility-potential < 0 [

```

```

    set responsibility-potential 0]
  if advancement-potential < 0 [
    set advancement-potential 0]
end

```

To initialize the workers, once created, the workers agents (termed “turtles” in NetLogo code) begin with a zeroed satisfaction and dissatisfaction score, representing a neutral state of both variables.

```

create-workers number-of-workers [
  setxy random-xcor random-ycor
  set color yellow
  set shape "person"
]
ask turtles [
  move-to patch-here
  set satisfaction 0
  set dissatisfaction 0

```

They are randomly allocated a designation for short- or long-term motivation factor preferences for use only in experiments that require the population be one of the two. They are also randomly allocated a designation of a singular motivation factor to prefer, which, again, is only used in experiments designated that this type of varied population be tested.

```

set halfinclination random 2
set fifthinclination random 5

```

Then the levels for motivation-style and hygiene-tolerance are set based on the distribution selected under motivation-factor-distribution.

```

if motivation-factor-distribution = "random" [
  set achievement-style random 10
  set recognition-style random 10
  set work-itself-style random 10
  set responsibility-style random 10
  set advancement-style random 10
  set policy-tolerance random 10
  set supervision-tolerance random 10
  set relationship-tolerance random 10
  set conditions-tolerance random 10
  set salary-tolerance random 10
]
if motivation-factor-distribution = "normal" [
  set achievement-style random-normal 5 2.5
  set recognition-style random-normal 5 2.5

```

```

        set work-itself-style random-normal 5 2.5
        set responsibility-style random-normal 5 2.5
        set advancement-style random-normal 5 2.5
        set policy-tolerance random-normal 5 2.5
        set supervision-tolerance random-normal 5 2.5
        set relationship-tolerance random-normal 5 2.5
        set conditions-tolerance random-normal 5 2.5
        set salary-tolerance random-normal 5 2.5
        reset-negatives-turtles
    ]
]
reset-ticks
end

to reset-negatives-turtles
    if achievement-style < 0 [
        set achievement-style 0]
    if recognition-style < 0 [
        set recognition-style 0]
    if work-itself-style < 0 [
        set work-itself-style 0]
    if responsibility-style < 0 [
        set responsibility-style 0]
    if advancement-style < 0 [
        set advancement-style 0]
    if policy-tolerance < 0 [
        set policy-tolerance 0]
    if supervision-tolerance < 0 [
        set supervision-tolerance 0]
    if relationship-tolerance < 0 [
        set relationship-tolerance 0]
    if conditions-tolerance < 0 [
        set conditions-tolerance 0]
    if salary-tolerance < 0 [
        set salary-tolerance 0]
end

```

4.2 Sub-models

The actual operation of the simulation is conducted in four procedures. In the first, the workers who have reached the maximum dissatisfaction score (100) leave the organization entirely and a new worker is hired in their place. Then, in the assess-hygiene procedure, the workers assess whether their hygiene tolerance levels have been exceeded, whether their motivation styles can be supported, and whether they need to move work units due to either. Third is the assess-state procedure where the worker calculates their satisfaction and dissatisfaction based on their final work unit position in the time-step. The final steps are the monitors of work unit populations and worker population status.

```

to go
  set step-die-count 0
  ask turtles [
    set moved? FALSE
    set firstmoved? FALSE
    set secondmoved? FALSE
    hire
    assess-hygiene
    assess-state
    count-workunit
    recolor-hygiene
  ]
  my-update-plots
  tick
end

```

The hire procedure includes worker departures and the hire of new replacements. The new hires are initialized in the same manner as the initial population.

```

to hire
  if dissatisfaction > 100 [

    ask patch-here [sprout-workers 1 [
      setxy random-xcor random-ycor
      set color yellow
      set shape "person"
      move-to patch-here
      set satisfaction 0
      set dissatisfaction 0
      set halfinclination random 2
      set fifthinclination random 5
      if motivation-factor-distribution = "random" [
        set achievement-style random 10
        set recognition-style random 10
        set work-itself-style random 10
        set responsibility-style random 10
        set advancement-style random 10
        set policy-tolerance random 10
        set supervision-tolerance random 10
        set relationship-tolerance random 10
        set conditions-tolerance random 10
        set salary-tolerance random 10
      ]
      if motivation-factor-distribution = "normal" [
        set achievement-style random-normal 5 2.5
        set recognition-style random-normal 5 2.5
        set work-itself-style random-normal 5 2.5
      ]
    ]
  ]
end

```



```

        set responsibility-style random-normal 5 2.5
        set advancement-style random-normal 5 2.5
        set policy-tolerance random-normal 5 2.5
        set supervision-tolerance random-normal 5 2.5
        set relationship-tolerance random-normal 5 2.5
        set conditions-tolerance random-normal 5 2.5
        set salary-tolerance random-normal 5 2.5
        reset-negatives-turtles
    ]

    ]

    ]
    set sprout-count (sprout-count + 1)
    set die-count (die-count + 1)
    set step-die-count (step-die-count + 1)
    die

    ]
end

```

The assess-hygiene procedure is conducted based on whether the workers are to calculate whether their hygiene tolerance has been exceeded in one of the hygiene factors (low-satisficing), two of the hygiene factors – one of which is always policy-tolerance (mid-satisficing), or all of the hygiene-factors (high). For all hygiene factors, except relationship-style, the worker is assessing their status vis-à-vis the style inherent in the work unit which remains constant. Relationship-style, however, is based on the average relationship-tolerances of the workers in the work unit.

This particular sub-model has the agents consider hygiene and motivation sequentially, rather than simultaneously. Sequential decision-tree design within agent-based modeling maintains computational parsimony and cognitive-plausibility and has been termed the “fast and frugal” decision making modeling method (Gigerenzer & Goldstein 1996; Kennedy, 2012). The fast and frugal method allows us to capture agents’ bounded rationality (Simon, 1997) and has been successfully used in simulating agents’ decision making in a variety of contexts, such as from agriculture, conflict, and evacuations (see Crooks et al., 2019 for a review).

```

to assess-hygiene
  if hygiene-weight = "low-satisficing" [
    (ifelse policy-tolerance > policy-style OR
      supervision-tolerance > supervision-style OR
      relationship-tolerance > relationship-style OR
      conditions-tolerance > conditions-style OR
      salary-tolerance > salary-style [
        assess-motivation
      ]
    ]
  ]
end

```

```

        [ first-move
      ])
    ]
    if hygiene-weight = "mid-satisficing" [
      (ifelse policy-tolerance > policy-style AND
        (supervision-tolerance > supervision-style OR
          relationship-tolerance > relationship-style OR
          conditions-tolerance > conditions-style OR
          salary-tolerance > salary-style) [
        assess-motivation
      ])
      [ first-move
    ])
  ]
  if hygiene-weight = "high" [
    (ifelse policy-tolerance > policy-style AND
      supervision-tolerance > supervision-style AND
      relationship-tolerance > relationship-style AND
      conditions-tolerance > conditions-style AND
      salary-tolerance > salary-style [
      assess-motivation
    ])
    [ first-move
  ])
]
end

```

If the worker's hygiene tolerance is not met, the agent moves to a neighboring patch. It is this proximity of move that mimics the bounded nature of what the worker knows is possible. Once the worker has moved, their satisfaction and dissatisfaction are reset to a neutral state. This is an explicit expression of the satisfaction/dissatisfaction being a representation of the person-job interaction.

```

to first-move
  if Org-size = "35x35" OR Org-size = "15x15" [
    rt random 50
    lt random 50
    fd 1
    move-to patch-here
    set moved? TRUE
    set firstmoved? TRUE
    set satisfaction 0
    set dissatisfaction 0
    assess-motivation
  ]
end

```

The worker then assesses their motivation. For some of the workers they are still in their same work unit as when they assessed their hygiene. For others, they have moved to a new work unit and have not assessed their hygiene prior to assessing their motivation. However, if their motivation needs are met, their hygiene in this work unit is assessed in the following timestep. If their motivation needs are not met, their move would have occurred regardless.

```
to assess-motivation
  if motivation-weight = "short-term" [
    if achievement-style > achievement-potential AND
      recognition-style > recognition-potential [
        second-move ]
  ]

  if motivation-weight = "long-term" [
    if work-itself-style > work-itself-potential AND
      responsibility-style > responsibility-potential AND
      advancement-style > advancement-potential [
        second-move ]
  ]

  if motivation-weight = "half-short-long" [
    if halfinclination = 0 [
      if achievement-style > achievement-potential AND
        recognition-style > recognition-potential [
          second-move ]
    ]
    if halfinclination = 1 [
      if work-itself-style > work-itself-potential AND
        responsibility-style > responsibility-potential AND
        advancement-style > advancement-potential [
          second-move ]
    ]
  ]

  if motivation-weight = "varied" [
    if fifthinclination = 0 [
      if achievement-style > achievement-potential [
        second-move ]
    ]
    if fifthinclination = 1 [
      if recognition-style > recognition-potential [
        second-move ]
    ]
    if fifthinclination = 2 [
      if work-itself-style > work-itself-potential [
```

```

        second-move ]
    ]
    if fifthinclination = 3 [
        if responsibility-style > responsibility-potential [
            second-move ]
        ]
    if fifthinclination = 4 [
        if advancement-style > advancement-potential [
            second-move ]
        ]
    ]
]

if motivation-weight = "all" [
    if achievement-style > achievement-potential AND
    recognition-style > recognition-potential AND
    work-itself-style > work-itself-potential AND
    responsibility-style > responsibility-potential AND
    advancement-style > advancement-potential [
        second-move ]
    ]
end

```

Their second move follows and ends the opportunities for moves.

to second-move ; turtle procedure

```

if Org-size = "35x35" OR Org-size = "15x15" [
    rt random 50
    lt random 50
    fd 1
    move-to patch-here
    set moved? TRUE
    set secondmoved? TRUE
    set satisfaction 0
    set dissatisfaction 0
]
end

```

Once the possible work unit moves are completed, the worker's satisfaction and dissatisfaction levels are calculated based on the weight the factor is empirically found to correlate with satisfaction or dissatisfaction. There are two empirically-calibrated options to test here. One is based on workforce satisfaction data (the analysis is detailed in Chapter 4) and called – GSA2013. The second is based on Herzberg's original 1959 study and called – Herzberg1959.

to assess-state

```

if calibration = "GSA2013" [

    if achievement-style < achievement-potential [
        set satisfaction satisfaction + 0.69 ]
    if recognition-style < recognition-potential [
        set satisfaction satisfaction + 0.59 ]
    if work-itself-style < work-itself-potential [
        set satisfaction satisfaction + 0.58 ]
    if responsibility-style < responsibility-potential [
        set satisfaction satisfaction + 0.43 ]
    if advancement-style < advancement-potential [
        set satisfaction satisfaction + 0.62 ]
    if policy-tolerance < policy-style [
        set dissatisfaction dissatisfaction + 0.63 ]
    if supervision-tolerance < supervision-style [
        set dissatisfaction dissatisfaction + 0.56 ]
    if relationship-tolerance < relationship-style [
        set dissatisfaction dissatisfaction + 0.57 ]
    if conditions-tolerance < conditions-style [
        set dissatisfaction dissatisfaction + 0.47 ]
    if salary-tolerance < salary-style [
        set dissatisfaction dissatisfaction + 0.45 ]
]
if calibration = "Herzberg1959" [
    if achievement-style < achievement-potential [
        set satisfaction satisfaction + 0.41 ]
    if recognition-style < recognition-potential [
        set satisfaction satisfaction + 0.33 ]
    if work-itself-style < work-itself-potential [
        set satisfaction satisfaction + 0.26 ]
    if responsibility-style < responsibility-potential [
        set satisfaction satisfaction + 0.23 ]
    if advancement-style < advancement-potential [
        set satisfaction satisfaction + 0.20 ]
    if policy-tolerance < policy-style [
        set dissatisfaction dissatisfaction + 0.31 ]
    if supervision-tolerance < supervision-style [
        set dissatisfaction dissatisfaction + 0.20 ]
    if relationship-tolerance < relationship-style [
        set dissatisfaction dissatisfaction + 0.15 ]
    if conditions-tolerance < conditions-style [
        set dissatisfaction dissatisfaction + 0.11 ]
    if salary-tolerance < salary-style [
        set dissatisfaction dissatisfaction + 0.17 ]
]

end

```

The key to the agent role in changing the environment is in the time-step changes to the work-unit's relationship-style. Here the relationship-style of the work-unit is changed to the mean relationship-tolerance of the workers in it.

```
to count-workunit
  set work-unit-population count turtles-here
  if work-unit-population > 0 [
    set relationship-style ((sum [relationship-tolerance] of
turtles-here) / (work-unit-population))
  ]
end
```

Bibliography

- Crooks, A., Malleson, N., Manley, E., & Heppenstall, A. (2019). *Agent-based Modelling and Geographical Information Systems: A Practical Primer*. Los Angeles: SAGE Publications.
- Grimm, Volker et al. 2006. "A Standard Protocol for Describing Individual-Based and Agent-Based Models." *Ecological Modelling*, 198(1): 115–26.
- Grimm, Volker et al. 2010. "The ODD Protocol: A Review and First Update." *Ecological Modelling*, 221(23): 2760–68.
- Gubler, T., Larkin, I., & Pierce, L. (2016). Motivational Spillovers from Awards: Crowding Out in a Multitasking Environment. *Organization Science*, 27(2), 286–303. <https://doi.org/10.1287/orsc.2016.1047>
- Herzberg, F. I., Mausner, B., & Snyderman, B. (1959). *The Motivation to Work* (2nd ed.). New York: John Wiley
- Lin, Z., Zhao, X., Ismail, K. M., & Carley, K. M. (2006). Organizational Design and Restructuring in Response to Crises: Lessons from Computational Modeling and Real-World Cases. *Organization Science; Linthicum*, 17(5), 598-618,672-673,675.
- Lundberg, C., Gudmundson, A., & Andersson, T. (2009). Herzberg's Two-Factor Theory of Work Motivation Tested Empirically on Seasonal Workers in Hospitality and Tourism. *Tourism Management*, 30(6), 890–899.
- Poulston, J. (2009). Working Conditions in Hospitality: Employees' Views of the Dissatisfactory Hygiene Factors. *Journal of Quality Assurance in Hospitality & Tourism*, 10(1), 23–43. <https://doi.org/10.1080/15280080902716993>
- Shapira, Z. (2011). "I've Got a Theory Paper—Do You?": Conceptual, Empirical, and Theoretical Contributions to Knowledge in the Organizational Sciences. *Organization Science*, 22(5), 1312–1321. <https://doi.org/10.1287/orsc.1100.0636>

Shirom A, Nirel N, Vinokur AD. (2010). Work hours and caseload as predictors of physician burnout: the mediating effects by perceived workload and by autonomy. *Appl Psychol.* 59(4):539–65.

Simon, H. A. (1955). A Behavioral Model of Rational Choice. *The Quarterly Journal of Economics*, 69(1), 99–118. JSTOR. <https://doi.org/10.2307/1884852>

Simon, H. (1969). *The Sciences of the Artificial*. MIT Press.

Waddimba, A. C., Scribani, M., Krupa, N., May, J. J., & Jenkins, P. (2016). Frequency of satisfaction and dissatisfaction with practice among rural-based, group-employed physicians and non-physician practitioners. *BMC Health Services Research*; London, 16. <http://dx.doi.org.mutex.gmu.edu/10.1186/s12913-016-1777-8>

Wilensky, U. (1999). NetLogo. <http://ccl.northwestern.edu/netlogo/>. Center for Connected Learning and Computer-Based Modeling, Northwestern University, Evanston, IL.